
1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc123632062]Corrections required
1.1 [bookmark: _Toc123632063]Correction to the Function fl_TC_9_1_10_1_TestBody
	Function name
	fl_TC_9_1_10_1_TestBody

	Reason for change
	1) In the current TTCN implementation, at Step 12 in the Registration Accept message, S-NSSAI with SST=3 is included in the Allowed NSSAI list, and the S-NSSAIs with SST=2 is included in the Configured NSSAI list.

In case the UE performs the PDU Session Establishment at Step 14a, according to current TTCN implementation the S-NSSAI with SST=1 is included in the PDU Session Establishment Accept.

This causes an anomaly at the UE side : the S-NSSAI present in Allowed list is NOT what is included in PDU Session establishment ACCEPT, and the UE releases the PDU Session causing testcase failure.
This needs to be addressed.

2) In current TTCN implementation, if the Step 14a is executed then at Step 18 in the Configuration Update Command mesaage the S-NSSAI with SST=3 also needs to be included along with S-NSSAI with SST=1. 
If this is not done, then the UE would release the PDU Session established earlier with S-NSSAI (SST=3). This also needs to be addressed.

3) In current TTCN implementation if the Step 14a is included, then at Step 22, the UE would include S-NSSAIs with SST=1 and SST=3 in the Registration REQ message. This needs to be accommodated.

	Summary of change
	1) At Step 14a, it is proposed to include S-NSSAI with SST=3 in the PDU Session Establishment Accept message. A Prose CR will be raised for this.
2) At Step 18, if pc_noOf_PDUsSameConnection >0 then the Allowed NSSAI is updated to include S-NSSAIs with SST=1 and SST=3. A Prose CR will be raised for this.
3) At Step 22, if pc_noOf_PDUsSameConnection >0 then the Requested NSSAI is updated to include S-NSSAIs with SST=1 and SST=3. A Prose CR will be raised for this.


	TTCN module
	NetworkSlicingEnh_NR5GC.ttcn

	MCC160 Comment
	



    
Before Change:
	<<SKIPPED CODE>>
//@siclog "Step 13" siclog@
    f_NR5GC_Registration_Complete(v_NGC_NASCellA);
    
    //@siclog "Step 14a" siclog@
    if (pc_noOf_PDUsSameConnection > 0 ){
      f_NR5GC_PDUSessionEstablishment (v_NGC_NASCellA, -, v_SRB);
      v_SRB := tsc_NR_RbId_SRB2; // SRB2 is established during PDU Session Establishment, hence further NAS signalling will happen on SRB2
    }
    v_SecurityParams := f_NR5GC_Security_Get();
    v_SecurityParams := f_NG_EAP_AuthenticationInit(v_SecurityParams, v_PLMN, hex2str(px_IMSI_Def)); // @sic R5s211584 sic@
    v_EAP := cs_EAP_Message (f_NG_EAP_ChallengeReq (v_SecurityParams, v_PLMN), omit);
    //@siclog "Step 15" siclog@
    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellA,
                                   v_SRB,
                                   cs_TimingInfo_Now,
                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                   cs_NG_NETWORK_SLICE_SPECIFIC_AUTHENTICATION_COMMAND(cs_S_NSSAI('01'O, '01'O, omit),v_EAP))));

    //@siclog "Step 16" siclog@
    SRB.receive (car_NR_SRB_NasPdu_IND(v_NGC_NASCellA, v_SRB, cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered,
                                                                      cr_NG_NETWORK_SLICE_SPECIFIC_AUTHENTICATION_COMPLETE(cs_S_NSSAI('01'O, '01'O, omit),?))))  -> value v_ReceivedAsp;

    if (not(f_NG_EAP_SuccessfulAuthenticationResponse(v_SecurityParams, v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.network_Slice_Specific_Authentication_Complete.eapMessage.eapMsg))){
    f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "EAP Authentication Failed");
    }

    //@siclog "Step 17" siclog@
    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellA,
                                   v_SRB,
                                   cs_TimingInfo_Now,
                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                   cs_NG_NETWORK_SLICE_SPECIFIC_AUTHENTICATION_RESULT(cs_S_NSSAI('01'O, '01'O, omit),cs_EAP_Message(bit2oct(encvalue(cs_EAP_Success)), omit)))));

    //@siclog "Step 18" siclog@
    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellA,
                                   v_SRB,
                                   cs_TimingInfo_Now,
                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                   cs_NG_CONFIGURATION_UPDATE_COMMAND(cs_ConfigUpdateInd('1'B, '1'B), -, -, v_AllowedNSSAI)))); // @sic R5-217939 sic@ // @sic R5s211584 sic@

    //@siclog "Step 19" siclog@
    SRB.receive (car_NR_SRB_NasPdu_IND(v_NGC_NASCellA, v_SRB, cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered,
                                                                      cr_NG_CONFIGURATION_UPDATE_COMPLETE)));

    //SS releases the RRC connection.
    //@siclog "Step 20" siclog@
    f_NR_RRCRelease(v_NGC_NASCellA);

    //@siclog "Step 21" siclog@
    //Void

    //@siclog "Step 22" siclog@
    //Check: Does the UE transmit a REGISTRATION REQUEST message with NSSAA set to '1'B ?
    v_ReceivedMsg := f_NR_RRC_ConnEst_DefWithNas (v_NGC_NASCellA, tsc_NR_RRC_TI_Def, ?);
    if (not f_Check_NG_RegistrationReqMsg (v_GMM_MobilityInfo, v_ReceivedMsg.Pdu, Mobility,-,-,-,-,cdr_NG_GMM_Cap_NSSAA,-,v_RequestedNSSAI)) {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Request Message Failed");
    }else {
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 22");
    }
<<SKIPPED CODE>>



After Change:
	<<SKIPPED CODE>>
//@siclog "Step 13" siclog@
    f_NR5GC_Registration_Complete(v_NGC_NASCellA);
    
    //@siclog "Step 14a" siclog@
    if (pc_noOf_PDUsSameConnection > 0 ) {
      //WA#9_1_10_1 COMMENTED OUT f_NR5GC_PDUSessionEstablishment (v_NGC_NASCellA, -, v_SRB);
      f_NR5GC_PDUSessionEstablishment (v_NGC_NASCellA, -, v_SRB, -, -, -, -, -, -, -, cs_S_NSSAI_WithIEI('22'O, cs_S_NSSAI('01'O, '03'O, omit))); //WA#9_1_10_1         
      v_SRB := tsc_NR_RbId_SRB2; // SRB2 is established during PDU Session Establishment, hence further NAS signalling will happen on SRB2
    }
    v_SecurityParams := f_NR5GC_Security_Get();
    v_SecurityParams := f_NG_EAP_AuthenticationInit(v_SecurityParams, v_PLMN, hex2str(px_IMSI_Def)); // @sic R5s211584 sic@
    v_EAP := cs_EAP_Message (f_NG_EAP_ChallengeReq (v_SecurityParams, v_PLMN), omit);
    //@siclog "Step 15" siclog@
    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellA,
                                   v_SRB,
                                   cs_TimingInfo_Now,
                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                   cs_NG_NETWORK_SLICE_SPECIFIC_AUTHENTICATION_COMMAND(cs_S_NSSAI('01'O, '01'O, omit),v_EAP))));

    //@siclog "Step 16" siclog@
    SRB.receive (car_NR_SRB_NasPdu_IND(v_NGC_NASCellA, v_SRB, cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered,
                                                                      cr_NG_NETWORK_SLICE_SPECIFIC_AUTHENTICATION_COMPLETE(cs_S_NSSAI('01'O, '01'O, omit),?))))  -> value v_ReceivedAsp;

    if (not(f_NG_EAP_SuccessfulAuthenticationResponse(v_SecurityParams, v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.network_Slice_Specific_Authentication_Complete.eapMessage.eapMsg))){
    f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "EAP Authentication Failed");
    }

    //@siclog "Step 17" siclog@
    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellA,
                                   v_SRB,
                                   cs_TimingInfo_Now,
                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                   cs_NG_NETWORK_SLICE_SPECIFIC_AUTHENTICATION_RESULT(cs_S_NSSAI('01'O, '01'O, omit),cs_EAP_Message(bit2oct(encvalue(cs_EAP_Success)), omit)))));

    //@siclog "Step 18" siclog@
    if (pc_noOf_PDUsSameConnection > 0 ){
    v_AllowedNSSAI := { //WA#9_1_10_1
      iei := '15'O,
      lvPart := {
        iel := '04'O,
        s_nssai := {cs_S_NSSAI('01'O, '03'O, omit), cs_S_NSSAI('01'O, '01'O, omit)}
      }};
    }    
    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellA,
                                   v_SRB,
                                   cs_TimingInfo_Now,
                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                   cs_NG_CONFIGURATION_UPDATE_COMMAND(cs_ConfigUpdateInd('1'B, '1'B), -, -, v_AllowedNSSAI)))); // @sic R5-217939 sic@ // @sic R5s211584 sic@
    

    //@siclog "Step 19" siclog@
    SRB.receive (car_NR_SRB_NasPdu_IND(v_NGC_NASCellA, v_SRB, cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered,
                                                                      cr_NG_CONFIGURATION_UPDATE_COMPLETE)));   
      
    //SS releases the RRC connection.
    //@siclog "Step 20" siclog@
    f_NR_RRCRelease(v_NGC_NASCellA);

    //@siclog "Step 21" siclog@
    //Void

    //@siclog "Step 22" siclog@
    //Check: Does the UE transmit a REGISTRATION REQUEST message with NSSAA set to '1'B ?
    if (pc_noOf_PDUsSameConnection > 0 ) {
        v_RequestedNSSAI := { //WA#9_1_10_1
          iei := '2F'O,
          lvPart := {
              iel := '04'O,
              s_nssai := {cs_S_NSSAI('01'O, '03'O, omit), cs_S_NSSAI('01'O, '01'O, omit)}
        }};
    }           
    v_ReceivedMsg := f_NR_RRC_ConnEst_DefWithNas (v_NGC_NASCellA, tsc_NR_RRC_TI_Def, ?);
    if (not f_Check_NG_RegistrationReqMsg (v_GMM_MobilityInfo, v_ReceivedMsg.Pdu, Mobility,-,-,-,-,cdr_NG_GMM_Cap_NSSAA,-,v_RequestedNSSAI)) {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Request Message Failed");
    }else {
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 22");
    }
<<SKIPPED CODE>>



